# Maintainer: David Runge <dvzrv@archlinux.org>

_name=hosc
pkgname=haskell-hosc
pkgver=0.20
pkgrel=9
pkgdesc="Haskell Open Sound Control"
arch=('loong64' 'x86_64')
url="https://github.com/rd--/hosc"
license=('GPL')
depends=('ghc-libs' 'haskell-blaze-builder' 'haskell-data-binary-ieee754' 'haskell-network')
makedepends=('ghc')
#source=("$pkgname-$pkgver.tar.gz::https://github.com/rd--/hosc/archive/${pkgver}.tar.gz")
source=("https://hackage.haskell.org/packages/archive/$_name/$pkgver/$_name-$pkgver.tar.gz")
sha512sums=('f04ea2cbb8a946c6aef246c4785518dd32f134c6d96643487402864d96a468244917091d36e8bf220ca64c5aab019ee859291354c3d8e50860d4169e9a6d0081')

prepare() {
  mv -v "${_name}-$pkgver" "$pkgname-$pkgver"
  cd "$pkgname-$pkgver"
}

build() {
  cd "$pkgname-$pkgver"
  runhaskell Setup configure -O \
                             --enable-shared \
                             --enable-executable-dynamic \
                             --disable-library-vanilla \
                             --prefix=/usr \
                             --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname \
                             --enable-tests \
                             --dynlibdir=/usr/lib \
                             --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
                             --datasubdir=$pkgname
  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd "$pkgname-$pkgver"
  runhaskell Setup test --show-details=direct
}

package() {
  cd "$pkgname-$pkgver"
  runhaskell Setup copy --destdir="${pkgdir}"
  install -vDm 744 register.sh "$pkgdir/usr/share/haskell/register/$pkgname.sh"
  install -vDm 744 unregister.sh "$pkgdir/usr/share/haskell/unregister/$pkgname.sh"
}
